﻿/*
* Copyright (c) 2006-2007 Erin Catto http://www.gphysics.com
*
* This software is provided 'as-is', without any express or implied
* warranty.  In no event will the authors be held liable for any damages
* arising from the use of this software.
* Permission is granted to anyone to use this software for any purpose,
* including commercial applications, and to alter it and redistribute it
* freely, subject to the following restrictions:
* 1. The origin of this software must not be misrepresented; you must not
* claim that you wrote the original software. If you use this software
* in a product, an acknowledgment in the product documentation would be
* appreciated but is not required.
* 2. Altered source versions must be plainly marked as such, and must not be
* misrepresented as being the original software.
* 3. This notice may not be removed or altered from any source distribution.
*/

package Box2D.Dynamics{


import Box2D.Collision.*;
import Box2D.Collision.Shapes.*;
import Box2D.Dynamics.Contacts.*;
import Box2D.Dynamics.*;
import Box2D.Common.Math.*;
import Box2D.Common.*;
import flash.events.IEventDispatcher;

import Box2D.Common.b2internal;
use namespace b2internal;

/**
 * Default implementation of IContactListener, which automatically raises
 * events on the world, and related bodies.
 */
public class b2DefaultContactListener implements IContactListener
{
	public function b2DefaultContactListener(world:b2World)
	{
		m_world = world;
	}
	
	public function BeginContact(contact:b2Contact):void
	{
		m_beginContact.contact = contact;
		m_world.dispatchEvent(m_beginContact);
		var dispatcher:IEventDispatcher;
		dispatcher = contact.m_fixtureA.m_body.m_eventDispatcher;
		if (dispatcher) dispatcher.dispatchEvent(m_beginContact);
		dispatcher = contact.m_fixtureB.m_body.m_eventDispatcher;
		if (dispatcher) dispatcher.dispatchEvent(m_beginContact);
	}
	public function EndContact(contact:b2Contact):void 
	{
		m_endContact.contact = contact;
		var dispatcher:IEventDispatcher;
		dispatcher = contact.m_fixtureA.m_body.m_eventDispatcher;
		if (dispatcher) dispatcher.dispatchEvent(m_endContact);
		dispatcher = contact.m_fixtureB.m_body.m_eventDispatcher;
		if (dispatcher) dispatcher.dispatchEvent(m_endContact);
		m_world.dispatchEvent(m_endContact);
	}
	public function PreSolve(contact:b2Contact, oldManifold:b2Manifold):void
	{
		m_preSolveContact.contact = contact;
		m_preSolveContact.oldManifold = oldManifold;
		m_world.dispatchEvent(m_beginContact);
		var dispatcher:IEventDispatcher;
		dispatcher = contact.m_fixtureA.m_body.m_eventDispatcher;
		if (dispatcher) dispatcher.dispatchEvent(m_preSolveContact);
		dispatcher = contact.m_fixtureB.m_body.m_eventDispatcher;
		if (dispatcher) dispatcher.dispatchEvent(m_preSolveContact);
	}
	public function PostSolve(contact:b2Contact, impulse:b2ContactImpulse):void
	{
		m_postSolveContact.contact = contact;
		m_postSolveContact.impulse = impulse;
		var dispatcher:IEventDispatcher;
		dispatcher = contact.m_fixtureA.m_body.m_eventDispatcher;
		if (dispatcher) dispatcher.dispatchEvent(m_postSolveContact);
		dispatcher = contact.m_fixtureB.m_body.m_eventDispatcher;
		if (dispatcher) dispatcher.dispatchEvent(m_postSolveContact);
		m_world.dispatchEvent(m_beginContact);
	}
	
	// Events for recycling
	private var m_beginContact:b2ContactEvent = new b2ContactEvent(b2World.BEGINCONTACT);
	private var m_endContact:b2ContactEvent = new b2ContactEvent(b2World.ENDCONTACT);
	private var m_preSolveContact:b2PreSolveEvent = new b2PreSolveEvent(b2World.PRESOLVE);
	private var m_postSolveContact:b2PostSolveEvent = new b2PostSolveEvent(b2World.POSTSOLVE);
	private var m_world:b2World;
}
	
	
}